WEATHER BUREAU
Office of Systems Development
Techniques Development Laboratory
Silver Spring, Md.

March 1969

An Operationally Oriented Small-Scale 500-Millibar Height Analysis Program

Technical Memorandum WBTM TDL 19

U.S. DEPARTMENT OF COMMERCE / ENVIRONMENTAL SCIENCE SERVICES ADMINISTRATION

ESSA TECHNICAL MEMORANDA

Weather Bureau, Techniques Development Laboratory Series

The primary purpose of the Techniques Development Laboratory of the Office of Systems Development is to translate increases in basic knowledge in meteorology and allied disciplines into improved operating techniques and procedures. To achieve this goal, TDL conducts and sponsors applied research and development aimed at improvement of diagnostic and prognostic methods for producing weather information. The laboratory carries out studies both for the general improvement of prediction methodology used in the National Meteorological Service System and for more effective utilization of weather forecasts by the ultimate user.

ESSA Technical Memoranda in the Weather Bureau Techniques Development Laboratory series facilitate rapid distribution of material which may be preliminary in nature and which may be published formally elsewhere at a later date. The first five papers in the TDL series are part of the former Weather Bureau Technical Notes series.

Papers listed below are available from the Clearinghouse for Federal Scientific and Technical Information, U.S. Department of Commerce, Sills Bldg., 5285 Port Royal Road, Springfield, Va. 22151. Price: $3.00 hard copy; $0.65 microfiche. Order by accession number shown in parentheses at end of each entry.

TN 10 TDL 1 Objective Prediction of Daily Surface Temperature. William H. Klein, Curtis W. Crockett, and Carlos R. Dunn, October 1965. (PB-168 590)

TN 11 TDL 2 Hurricane Cindy Galveston Bay Tides. N. A. Pore, A. T. Angelo, and J. G. Taylor, September 1965. (PB-168 608)

TN 29 TDL 3 Atmospheric Effects on Re-Entry Vehicle Dispersions. Karl R. Johannessen, December 1965. (PB-169 381)

TN 45 TDL 4 A Synoptic Climatology of Winter Precipitation from 700-mb. Lows for the Intermountain Areas of the West. D. L. Jorgensen, W. H. Klein, and A. F. Korte, May 1966. (PB-170 635)

TN 47 TDL 5 Hemispheric Specification of Sea Level Pressure from Numerical 700-mb. Height Forecasts. William H. Klein and Billy M. Lewis, June 1966. (PB-173 091)

WBTM TDL 6 A Fortran Program for the Calculation of Hourly Values of Astronomical Tide and Time and Height of High and Low Water. N. A. Pore and R. A. Cummings, January 1967. (PB-174 660)

WBTM TDL 7 Numerical Experiments Leading to the Design of Optimum Global Meteorological Networks. M. A. Alaka and F. Lewis, February 1967. (PB-174 497)

WBTM TDL 8 An Experiment in the Use of the Balance Equation in the Tropics. M. A. Alaka, D. T. Rubsam, and G. E. Fisher, March 1967. (PB-174 501)

WBTM TDL 9 A Survey of Studies of Aerological Network Requirements. M. A. Alaka, May 1967. (PB-174 984)

WBTM TDL 10 Objective Determination of Sea Level Pressure from Upper Level Heights. William Klein, Frank Lewis, and John Stackpole, May 1967. (PB-179 949)

WBTM TDL 11 Short Range, Subsynoptic Surface Weather Prediction. H. R. Glahn and D. A. Lowry, July 1967. (PB-175 772)

WBTM TDL 12 Charts Giving Station Precipitation in the Plateau States from 700-Mb. Lows During Winter. D. L. Jorgensen, A. F. Korte, and J. A. Bunce, Jr., October 1967. (PB-176 742)

WBTM TDL 13 Interim Report on Sea and Swell Forecasting. N. A. Pore and W. S. Richardson, December 1967. (PB-177 038)

WBTM TDL 14 Meteorological Analysis of 1964-65 ICAO Turbulence Data. DeVer Colson, September 1968.

WBTM TDL 15 Prediction of Temperature and Dew Point by Three-Dimensional Trajectories. Ronald M. Reap, September 1968.

(Continued on inside back cover)

U.S. DEPARTMENT OF COMMERCE
Environmental Science Services Administration
Weather Bureau

ESSA Technical Memorandum WBTM TDL 19

AN OPERATIONALLY ORIENTED SMALL-SCALE 500-MILLIBAR HEIGHT ANALYSIS PROGRAM

Harry R. Glahn and George W. Hollenbaugh

OFFICE OF SYSTEMS DEVELOPMENT
TECHNIQUES DEVELOPMENT LABORATORY

SILVER SPRING, MD.
March 1969

AN OPERATIONALLY ORIENTED SMALL-SCALE 500-MB HEIGHT ANALYSIS PROGRAM

Harry R. Glahn and George W. Hollenbaugh

ABSTRACT

A computer program which performs analyses of 500-mb heights over the relatively dense data region of the United States and Canada by the method of successive approximations is described. The program includes error-detecting routines and can be used in an operational environment. The detail specified by the radiosonde reports is retained with a gridlength of about 100 miles. A listing of the program is included.

INTRODUCTION

In support of a task to investigate the possible advantages of reducing the gridlength used in numerical prediction models, a computer program was developed to perform small-scale 500-mb height analyses over the relatively dense data network of the United States and Canada. The method used is basically that of Bergthórssen and Döös [1] and Cressman [2]; however, differences in details do exist. Error-detecting routines are included, and observed winds are used in implying height gradients in a geostrophic manner.

ANALYSIS AREA AND DATA

The analysis area, shown in Figure 1., is covered by a 35 x 35 grid which has a gridlength of about 100 miles, exactly ½ that normally used at the National Meteorological Center (NMC). Alternate gridpoints in the "half grid" are gridpoints in the NMC "full grid". The upper air reporting stations are also shown in Fig. 1; the station spacing is somewhat greater than the grid spacing even in the most dense data areas.

The data used are the 500-mb height and wind reports from the radiosonde network which are collected on magnetic tape, decoded, and checked for hydrostatic consistency at NMC.

THE ANALYSIS METHOD

The basic analysis method used at NMC is still that described by Cressman [2] in 1959, although some of the details have undergone change.* This method is one of successive approximations and allows the use of the geostrophic or some other wind-height relationship so that reported winds can be used together with reported heights in adjusting the required "first guess" height analysis.

---

* McDonell, J. E., "Notes on NMC Objective Analysis Programs" and attachments, unpublished manuscript, March, 1968.



Figure 1. The Analysis area and superimposed 35 x 35 grid. The observing stations are shown by black dots.

The analysis is performed in a series of cycles or passes over the data. In most applications the first guess may be considerably in error, and 3 or 4 passes are necessary. However, in this application, the first guess is the analysis performed by NMC on the full grid and interpolated to the half grid. Therefore, the first guess is very good, and only 2 scans are necessary to insert the detail possible with the half grid.

On each pass, an observation occasions a change or correction to all gridpoints within a distance of R gridlengths, where R is called the scan radius or radius of influence. R varies with pass number and usually decreases for later passes.

A correction term is computed based on each observation. This correction is one of three types:

$$C_{1_{i,j}} = (Z_{x,y} - \hat{Z}_{x,y}) \qquad \text{if only height is available}$$

$$C_{2_{i,j}} = [Z_{x,y} + (i - x)\frac{\partial Z}{\partial x} + (j - y)\frac{\partial Z}{\partial y} - \hat{Z}_{i,j}] \qquad \text{if both height and wind are available}$$

$$C_{3_{i,j}} = [\hat{Z}_{x,y} + (i - x)\frac{\partial Z}{\partial x} + (j - y)\frac{\partial Z}{\partial y} - \hat{Z}_{i,j}] \qquad \text{if only wind is available}$$

where

$Z_{x,y}$ = observed height at station at position x,y

$\hat{\hat{Z}}_{x,y}$ = height at station interpolated from current analysis

$\frac{\partial Z}{\partial x}, \frac{\partial Z}{\partial y}$ = height gradients in x and y directions respectively

$\tilde{Z}_{i,j}$ = height at gridpoint i,j from current analysis

The height gradients are computed by the geostrophic relationships:

$$\frac{\partial Z}{\partial x} = \frac{fV}{g} = .782 \sin \Phi(1 + \sin\Phi)V$$

$$\frac{\partial Z}{\partial y} = -\frac{fU}{g} = - .782 \sin \Phi(1 + \sin\Phi)U$$

where U and V are the observed eastward and northward wind components, respectively, Φ is the latitude of the observing station, g is the gravitational acceleration, and f the Coriolis parameter. When U and V are used in knots, $\frac{\partial Z}{\partial x}$ and $\frac{\partial Z}{\partial y}$ will be in meters/grid unit at the latitude of the station.

The individual correction terms are then combined into a total correction for the gridpoint in one of three ways:

$$C_a = \frac{1}{n}\sum_{k=1}^{n} C_{mk}$$

$$C_b = \frac{1}{n}\sum_{k=1}^{n} \frac{R^2-d_k^2}{R^2+d_k^2} C_{mk}$$

$$C_c = \frac{\frac{1}{n}\sum_{k=1}^{n} \frac{R^2-d_k^2}{R^2+d_k^2} C_{mk}}{\frac{1}{n}\sum_{k=1}^{n} \frac{R^2-d_k^2}{R^2+d_k^2}}$$

Where n = number of observations within R of the gridpoint,

d_k = the distance from the observation to the gridpoint,

m = the type of correction 1, 2, or 3, and

C_{mk} = the individual correction terms.

The type of correction to make for each observation is determined by the program; a height or wind is available only if reported and acceptable by the error-detection routine. The way the corrections are combined is controlled by a word on a control card.

C_b and C_c perform a variable weighting of the individual corrections; the closer a station is to the gridpoint, the greater is its weight. The effect of using C_a or C_c rather than C_b is a greater change being made at the gridpoint and more rapid convergence to the general level of the data within the influence circle.

ERROR-DETECTING PROCEDURE

Errors in the observations are detected by comparing them to the current analysis (first guess for the first pass). The height is judged to be in error and not used in the analysis if

$$|Z_{x,y} - \hat{Z}_{x,y}| > ERZ$$

where ERZ is specified by control card. The wind is judged to be in error and not used if

$$|D_{x,y} - \hat{D}_{x,y}| > \text{ERD} \qquad \text{and} \qquad V_{x,y} \geq 15$$

or

$$|V_{x,y} - \hat{V}_{x,y}| > \text{ERS}$$

where D and V are the direction and speed of the wind, respectively, and the carat indicates interpolated values. The direction difference is measured so as to be not greater than 180°. A wind will not be discarded because of its direction unless its speed is $\geq$ 15 kts. ERD and ERS are specified by control card.

SMOOTHING

Usually some type of smoothing is done after one or more of the data passes. The smoothing operator employed in this program is that used by the Travelers Research Center [3] and is a generalization of the one given by Cressman [2]:

$$S_{i,j} = \frac{A_{i,j} + b\bar{A}_{i,j}}{1 + b}$$

where

$$\bar{A}_{i,j} = \tfrac{1}{4}(A_{i+1,j} + A_{i,j+1} + A_{i,j-1} + A_{i-1,j})$$

and $S_{i,j}$ is the smoothed gridpoint value. Specific values of b used are given in the following section.

APPLICATION OF THE ANALYSIS PROGRAM

The first guess is arrived at by interpolating with Bessel's formula from the NMC full grid analysis. An example of a contoured first guess field is shown in Fig. 2. A specific combination of type of correction and values of R, ERZ, ERD, ERS, and b which gives good results is shown in Table 1.

Table 1. A combination of program parameters which gives good results

Pass No.	R(Grid Units)	ERZ (Meters)	ERD (Degrees)	ERS (Knots)	Type of Correction	b
1	3	35	35	25	C_c	2
2	2	35	35	25	C_b	1



Figure 2. First guess 500-mb height field for May 19, 1967, 0000 GMT, interpolated from NMC analysis.

Combinations of parameters other than that shown in Table 1 will also give good results, and repeated use of the program might suggest a better set. If the acceptance criteria ERZ, ERD, and ERS are lowered, more data will be discarded; an erroneous value will more likely be thrown out, but also some good data will be judged incorrect.

Figure 3 shows the final analysis after two data passes for the same case as shown in Figure 2. The data on which the analysis is based are shown in Figure 4, with the final analysis traced from Figure 3.

The small-scale analysis and the interpolated NMC analysis are, of course, much the same. More detail is shown in the small-scale analysis in a few areas:

(1) The low center near 55N and 95W (see Fig. 4 for location) on the small-scale analysis is lower and slightly to the northwest of the low on the full-grid analysis

(2) The low center near 65N and 80W is deeper on the small-scale analysis

(3) The low center in eastern Quebec is more distinguishable from the trough to the northeast on the TDL analysis

(4) The ridge from the northeast extends farther into the Labrador-Newfoundland area on the half-grid analysis.

In each case the detail seems justified and the data are fit very nicely. Only two data within the 35 x 35 grid area were judged incorrect and not used by the program--the wind in north-central Mexico (near 29N, 106W) and the wind at Caribou, Maine. In the former case, the direction appeared to be incorrect, and in the latter, the speed seemed to be too low. Data outside the analysis area but plotted in Fig. 4 were included in making the analysis.

Mean-square and mean absolute differences were computed between both the observed wind speeds and heights and the wind speeds and heights interpolated from the NMC and TDL analyses. The NMC analysis had already been interpolated by Bessel's formula to the TDL half grid. Then interpolation to station locations was also done with Bessel's formula. Geostrophic winds were calculated at each grid point before interpolation. The results are given in Table 2.



Figure 3. Final 500-mb height analysis for May 19, 1967, 0000 GMT.



Figure 4. Observed data and final analysis traced from Figure 3.

Table 2. Mean-square (MSD) and mean absolute differences (MAD) in wind speed in m/sec. and height in meters between observations and analyses

Analysis	Wind Speed			Height		
	No. of Stations	MAD	MSD	No. of Stations	MAD	MSD
NMC	93	4.9	6.3	96	6.6	8.7
TDL	93	4.1	5.5	96	3.4	4.9

All of the reported winds within the grid were used in computing the differences. The observations contain some error and the computed differences are partly due to these errors. Also, the observations can be fit too closely, resulting in a ragged analysis. This ragged analysis would not be desirable and statistics like those shown in Table 2 might be misinterpreted. However, it is felt the TDL analysis is of about the desired smoothness and that the statistics in Table 2 indicate a better analysis is possible with a grid length of about 100 miles than with a grid length of twice that. This conclusion is based on several such comparisons.

The geostrophic wind relationship is used in this program. Considerable effort was expended in an attempt to produce a better analysis with an approximation to the gradient wind relationship. However, no overall improvement was noted, and, in fact, statistics like those in Table 2 indicated the analysis with the gradient wind approximation to be less desirable.

REFERENCES

1. Bergthórssen, P. and B. R. Döös, "Numerical Weather Map Analysis", <u>Tellus</u>, Vol. 7, No. 3, August, 1955, pp. 329-340.

2. Cressman, G. P., "An Operational Objective Analysis System", <u>Monthly Weather Review</u>, Vol. 87, No. 10, October, 1959, pp. 367-374.

3. Thomasell, A., Jr., and J. G. Welsh, "Objective Analysis of Sea Level Pressure and Surface Temperature, Dew Point and Wind", The Travelers Research Center, Inc., <u>Tech. Publication</u> 19, Contract FAA/BRD-363, November, 1962, 87 pp.

APPENDIX

This appendix contains the listings of two subroutines in FORTRAN language necessary for performing the analysis described. These subroutines are in the form for use with the SCOPE Operating System on the CDC 6600 at Suitland, Maryland.

Subroutine ANALYZ is called when an analysis is desired and it, in turn, uses NTRP for interpolation. Comment cards describe the program parameters.

Upon entry to ANALYZ, certain information must be present in COMMON:

ADP (150,9): Basic data, and computed variables described in comment cards

T(35,35) : First guess field of 500-mb heights in meters

NADP : Number of datum values in ADP, that is, the number of rows occupied.

The lower left corner of the grid is considered to be position (1,1) with the first subscript increasing upward to 35 and the second subscript increasing to the right to 35.

The subroutine OUTPT called by ANALYZ is used for outputting maps. The calling statement can be eliminated or another subroutine substituted.

```
      SUBROUTINE ANALYZ
C        D523
C        JUNE, 1968    HOLLENBAUGH
C        TO MAKE A 500 MB ANALYSIS BY THE BERGTHORSSEN-CRESSMAN-DOOS
C            (BCD) METHOD
C        SUBROUTINE ARGUMENTS AND CONTROL INFORMATION
C                  ERZ=MAX ERROR FROM LAST PASS (OR FIRST GUESS) HEIGHT
C            IN METERS
C                  ERD=MAX ERROR FROM LAST PASS (OR 1ST GUESS) WIND DIR.
C            IN DEGREES
C                  ERS=MAX ERROR FROM LAST PASS (OR 1ST GUESS) WIND SPD.
C            IN KNOTS
C                  RDS=RADIUS OF INFLUENCE IN GRID UNITS
C                  SMH=SMOOTHING PARAMETER (NO SMOOTHING DONE IF SMH=0)
C                  KRT=CORRECTION TYPE
C                      1 MEANS W=1
C                      2 MEANS W=(RSQ-DSQR)/(RSQ+DSQR)
C                      3 MEANS SAME AS TYPE 2 EXCEPT SUM OF WTS IN DEN.
C                  NPS=NUMBER OF PASSES
C                 NADP=HOLDS NUMBER OF DATA POINTS
C               T( , )=HOLDS CURRENT ANALYSIS
C             ADP(N,1)=HOLDS STATION DATA X-POSITION
C             ADP(N,2)=HOLDS STATION DATA Y-POSITION
C             ADP(N,3)=HOLDS STATION DATA HEIGHT IN METERS
C             ADP(N,4)=HOLDS STATION DATA WIND DIRECTION IN DEGREES
C             ADP(N,5)=HOLDS STATION DATA WIND SPEED IN KNOTS
C             ADP(N,6)=HOLDS STATION DATA GRID ORIENTED WIND DIRECTION
C                      IN VECTOR NOTATION MEASURED CLOCKWISE FROM NORTH
C                      IN RADIANS
C             ADP(N,7)=HOLDS STATION DATA GRID ORIENTED WIND GRADIENT
C                      SCALING FACTOR IN M/GRID INTERVAL
C             ADP(N,8)=HOLDS STATION DATA WIND GRADIENT (DZ/DX) IN X
C                      DIRECTION IN M/GRID INTERVAL
C             ADP(N,9)=HOLDS STATION DATA WIND GRADIENT (DZ/DY) IN Y
C                      DIRECTION IN M/GRID INTERVAL
C             KILH(N) =0 IF STATION HEIGHT IS JUDGED TO BE CORRECT AND
C                      IS USED IN ANALYSIS, OTHERWISE = 1
C             KILW(N) =0 IF STATION WIND IS JUDGED TO BE CORRECT AND
C                      IS USED IN ANALYSIS, OTHERWISE = 1
C               DX( , )=DZ/DX IN M/GRID INTERVAL
C               DY( , )=DZ/DY IN M/GRID INTERVAL
C             CORR( , )=HOLDS CORRECTIONS TO BE MADE ON CURRENT PASS
C             CONT( , )=HOLDS NUMBER OF STATIONS AFFECTING GRID POINT
C                      ON CURRENT PASS
C        A MISSING DATUM IS INDICATED IN VARIOUS ARRAYS BY 9999.
      DIMENSION ADP(150,9),T(35,35),CORR(35,35),CONT(35,35),
     1 KILH(150),KILW(150),VALZ(150),VALX(150),VALY(150),
     2 DX(35,35),DY(35,35)
      COMMON/BLOCK3/ADP,NADP
      COMMON/BLOCK12/T
C         ******************************************************
      READ 110,ERZ,ERD,ERS,RDS,SMH,KRT,NPS
  110 FORMAT(5F4.0,2I4)
      PRINT 112
```

```
 112   FORMAT(1H1)
       DO 480 LP=1,NPS
       IF(LP-1)125,125,115
 115   READ 120,ERZ,ERD,ERS,RDS,SMH,KRT
 120   FORMAT(5F4.0,I4)
 125   RSQ=RDS*RDS
       PRINT 128,LP,RDS,KRT,SMH,ERZ,ERD,ERS
 128   FORMAT(///,I4,F5.1,I4,4F7.1,/)
       ERD=.01745329252*ERD
C         CHANGES ERD FROM DEGREES TO RADIANS
       DO 130 J=1,35
       DO 130 I=1,35
       CORR(J,I)=CONT(J,I)=0.
 130   CONTINUE
       DO 131 J=2,34
       DO 131 I=2,34
       DX(J,I)=(T(J,I+1)-T(J,I-1))/2.
       DY(J,I)=(T(J+1,I)-T(J-1,I))/2.
 131   CONTINUE
       DO 132 I=2,34
       DX(1,I)=DX(2,I)
       DY(1,I)=DY(2,I)
       DX(35,I)=DX(34,I)
       DY(35,I)=DY(34,I)
 132   CONTINUE
       DO 133 J=1,35
       DX(J,1)=DX(J,2)
       DY(J,1)=DY(J,2)
       DX(J,35)=DX(J,34)
       DY(J,35)=DY(J,34)
 133   CONTINUE
       DO 245 K=1,NADP
       KILH(K)=KILW(K)=0
       IF(1.-RDS-ADP(K,1))135,135,150
 135   IF(RDS+35.-ADP(K,1))150,140,140
 140   IF(1.-RDS-ADP(K,2))145,145,150
 145   IF(RDS+35.-ADP(K,2))150,155,155
 150   VALZ(K)=VALX(K)=VALY(K)=9999.
       KILH(K)=KILW(K)=1
       GO TO 245
 155   CALL NTRP(T,ADP(K,2),ADP(K,1),VALZ(K))
       IF(ADP(K,3)-9999.)160,175,160
 160   IF(ABSF(ADP(K,3)-VALZ(K))-ERZ)180,180,165
 165   PRINT 170,ADP(K,3),VALZ(K),ADP(K,1),ADP(K,2),LP
 170   FORMAT(*0STATION REPORTED HEIGHT*,F9.2,* NOT ACCEPTED, INCONSISTEN
      1T WITH ANALYSIS*,F9.2,* COORDINATES*,2F7.2,* PASS*,I3)
 175   KILH(K)=1
 180   IF(ADP(K,6)-9999.)185,240,185
 185   CALL NTRP(DX,ADP(K,2),ADP(K,1),VALX(K))
       CALL NTRP(DY,ADP(K,2),ADP(K,1),VALY(K))
       IF(VALX(K))195,190,195
 190   TANTH=-VALY(K)*999999.
       GO TO 200
 195   TANTH=-VALY(K)/VALX(K)
```

```
200   EPHI=ATANF(TANTH)
      IF(EPHI)202,208,208
202   IF(VALX(K))204,206,206
204   EPHI=EPHI+3.14159265
      GO TO 210
206   EPHI=EPHI+6.28318530
      GO TO 210
208   IF(VALX(K))204,210,210
210   ADIFF=ABSF(ADP(K,6)-EPHI)
      IF(ADIFF-3.14159265)212,212,211
211   ADIFF=6.28318530-ADIFF
212   IF(ADIFF-ERD)225,225,213
213   IF(ADP(K,5)-15.)225,215,215
215   PRINT 220,ADP(K,6),EPHI,ADP(K,1),ADP(K,2),LP
220   FORMAT(*0STATION WIND DIRECTION*,F10.2,* NOT ACCEPTED, INCONSISTEN
     1T WITH ANALYSIS*,F9.2,* COORDINATES*,2F7.2,* PASS*,I3)
      GO TO 240
225   ESPD=(SQRTF(VALX(K)**2+VALY(K)**2))/ADP(K,7)
      IF(ABSF(ADP(K,5)-ESPD)-ERS)245,245,230
230   PRINT 235,ADP(K,5),ESPD,ADP(K,1),ADP(K,2),LP
235   FORMAT(*0STATION WIND SPEED*,F14.2,* NOT ACCEPTED, INCONSISTENT WI
     1TH ANALYSIS*,F9.2,* COORDINATES*,2F7.2,* PASS*,I3)
240   KILW(K)=1
245   CONTINUE
      DO 380 K=1,NADP
      IF(KILH(K).EQ.1.AND.KILW(K).EQ.1)380,260
260   IMIN=ADP(K,1)-RDS+.999
      IF(IMIN)265,265,270
265   IMIN=1
      GO TO 275
270   IF(IMIN-35)275,275,380
275   IMAX=ADP(K,1)+RDS+.001
      IF(IMAX)380,380,280
280   IF(IMAX-35)290,290,285
285   IMAX=35
290   JMIN=ADP(K,2)-RDS+.999
      IF(JMIN)295,295,300
295   JMIN=1
      GO TO 305
300   IF(JMIN-35)305,305,380
305   JMAX=ADP(K,2)+RDS+.001
      IF(JMAX)380,380,310
310   IF(JMAX-35)320,320,315
315   JMAX=35
320   DO 375 J=JMIN,JMAX
      DO 375 I=IMIN,IMAX
      DSQR=(FLOATF(I)-ADP(K,1))**2+(FLOATF(J)-ADP(K,2))**2
      IF(DSQR-RSQ)325,325,375
325   IF(KILH(K).EQ.0.AND.KILW(K).EQ.0)330,335
330   EHGT=ADP(K,3)+(I-ADP(K,1))*ADP(K,8)+(J-ADP(K,2))*ADP(K,9)-T(J,I)
      GO TO 350
335   IF(KILH(K).EQ.1.AND.KILW(K).EQ.0.)340,345
340   EHGT=VALZ(K)+(I-ADP(K,1))*ADP(K,8)+(J-ADP(K,2))*ADP(K,9)-T(J,I)
      GO TO 350
```

```
345   EHGT=ADP(K,3)-VALZ(K)
350   IF(KRT-1)355,355,365
355   CORR(J,I)=CORR(J,I)+EHGT
360   CONT(J,I)=CONT(J,I)+1.
      GO TO 375
365   WT=(RSQ-DSQR)/(RSQ+DSQR)
      CORR(J,I)=CORR(J,I)+WT*EHGT
      IF(KRT-2)360,360,370
370   CONT(J,I)=CONT(J,I)+WT
375   CONTINUE
380   CONTINUE
      DO 405 J=1,35
      DO 405 I=1,35
      IF(CONT(J,I))405,405,400
400   T(J,I)=T(J,I)+CORR(J,I)/CONT(J,I)
405   CONTINUE
      CALL OUTPT(T,1.,0.,0,40,0,LP*2+1)
      IF(SMH)480,480,425
425   SM=SMH+1.
      DO 470 J=1,35
      DO 470 I=1,35
      SUM=0.
      KT=0
      K5=I-1
      IF(K5)435,435,430
430   KT=KT+1
      SUM=SUM+T(J,K5)
435   K5=I+1
      IF(K5-35)440,440,445
440   KT=KT+1
      SUM=SUM+T(J,K5)
445   K5=J-1
      IF(K5)455,455,450
450   KT=KT+1
      SUM=SUM+T(K5,I)
455   K5=J+1
      IF(K5-35)460,460,465
460   KT=KT+1
      SUM=SUM+T(K5,I)
465   CORR(J,I)=(T(J,I)+SMH*SUM/FLOATF(KT))/SM
470   CONTINUE
      DO 475 J=1,35
      DO 475 I=1,35
      T(J,I)=CORR(J,I)
475   CONTINUE
      CALL OUTPT(T,1.,0.,0,40,0,LP*2+2)
480   CONTINUE
      RETURN
      END
```

```
      SUBROUTINE NTRP(P,BY,BX,BB)
         D521
         JUNE, 1968    HOLLENBAUGH
         PERFORMS BI-QUADRATIC INTERPOLATION WHERE POSSIBLE, LINEAR
            INTERPOLATION IN OUTSIDE GRID INTERVAL, AND LINEAR
            EXTRAPOLATION OUTSIDE GRID
         ARGUMENTS
            P( , )=ARRAY FROM WHICH INTERPOLATION IS MADE
            BY=Y-COORDINATE, FROM BOTTOM
            BX=X-COORDINATE, FROM LEFT
            BB=INTERPOLATED (OR EXTRAPOLATED) VALUE
      DIMENSION P(35,35),B(4)
        *********************************************************
104   NBX=BX
      NBY=BY
      IF(NBX-1)114,120,111
111   IF(NBX-34)112,120,115
112   IF(NBY-1)121,130,113
113   IF(NBY-34)140,130,123
114   NBX=1
      GO TO 120
115   NBX=34
120   IF(NBY-1)121,130,122
121   NBY=1
      GO TO 130
122   IF(NBY-35)130,123,123
123   NBY=34
         STATEMENT 130 STARTS BI-LINEAR INTERPOLATION-EXTRAPOLATION
130   NBXP1=NBX+1
      NBYP1=NBY+1
      DX=BX-FLOATF(NBX)
      DY=BY-FLOATF(NBY)
      BB=P(NBY,NBX)+(P(NBY,NBXP1)-P(NBY,NBX))*DX+(P(NBYP1,NBX)-
     1P(NBY,NBX))*DY+(P(NBY,NBX)+P(NBYP1,NBXP1)-P(NBYP1,NBX)-P(NBY,
     2NBXP1))*DX*DY
      RETURN
         STATEMENT 140 STARTS BI-QUADRATIC INTERPOLATION
140   DX=BX-FLOATF(NBX)
      DY=BY-FLOATF(NBY)
      NBYP2=NBY+2
      NBYP1=NBY+1
      NBYM1=NBY-1
      FCT=(DY**2-DY)/4.
      FET=(DX**2-DX)/4.
      DO 145 J=1,4
      N=NBX-2+J
      B(J)=P(NBY,N)+(P(NBYP1,N)-P(NBY,N))*DY+(P(NBYM1,N)+P(NBYP2,N)-
     1P(NBY,N)-P(NBYP1,N))*FCT
145   CONTINUE
      BB=B(2)+(B(3)-B(2))*DX+(B(1)+B(4)-B(2)-B(3))*FET
      RETURN
      END
```

(Continued from inside front cover)

WBTM TDL 16 Objective Visibility Forecasting Techniques Based on Surface and Tower Observations. Donald M. Gales, October 1968. (PB-180 479)

WBTM TDL 17 Second Interim Report on Sea and Swell Forecasting. N. A. Pore and Lt. W. S. Richardson, USESSA, January 1969. (PB-182 273)

WBTM TDL 18 Conditional Probabilities of Precipitation Amounts in the Conterminous United States. Donald L. Jorgensen, William H. Klein, and Charles F. Roberts, March 1969. (PB-183 144)